
A Brazilian Portuguese work on MetaPost,
and how mathematics is embedded in it

Estevão Vińıcius Candia

Abstract

This article presents a summary of my final work
[1] for the master’s degree course in Mathematics,
presented on November 1, 2018. This work shows an
introduction to the basic commands of the MetaPost
language. For this, some mathematical concepts are
explained, as they are part of its programming. It
is shown how these concepts can be used to create
high-quality figures used in areas of mathematics
such as geometry and the study of graphs of real
functions.

Introduction

Written communication is the basis for teaching and
learning in various areas of knowledge. Writing a
legible and elegant mathematical work was not so
simple in the old days without the computational
tools available today. The ability to expose knowl-
edge in accordance with technical requirements, or
even personal norms, moved people to create typo-
graphic writing systems that included mathematical
symbols and concepts. MetaPost is a very useful tool
for this.

When I needed a research topic for my master’s
thesis, my advisor, Professor Elisabete Freitas, in-
troduced me to MetaPost and said that it would be
very nice to study some of the mathematics behind
it. She said she uses MetaPost whenever she needs
to create pictures for her work on TEX and that it
has powerful tools that can contribute to the work of
Mathematics teachers. So, I met this language and
started studying its pedagogical use.

What did I study?

Chapter 1 of the dissertation presents some com-
mands for creating basic figures for an interested
reader who has never heard of MetaPost. To iden-
tify the positions in a drawing, MetaPost uses the
one-to-one correspondence between the points of a
plane and the set of ordered pairs of real numbers.
Mathematics!

In this initial chapter, the reader learns about
the commands draw, drawdot, cycle, paths with --

and .., style commands that change the width, size,
or shape of a solid or dashed line, label, dotlabel,
drawarrow, drawdblarrow, some data types, com-
mands for colors, filling and mathematical operators.
It also contains the conversion table of the units of
measurement used in MetaPost, as shown in Table

1 and a short tutorial on how to insert a MetaPost
figure into a TEX document.

Unit of measurement Conversion
pt 0,035145 cm
bp 0,035278 cm
in 2,54 cm
pc 0,423333 cm
mm 0,1 cm
cc 0,451167 cm
dd 0,0376 cm

Table 1: Conversion table of the units of measurement
used in MetaPost

Some mathematical concepts of vectors and ge-
ometry can be used to make drawings in MetaPost.
MetaPost has vector operators and is able to under-
stand the data of the type pair as vectors. Chapter 2
addresses the mathematical basis for this, as well as
some applications of these operations in geometry.

This chapter defines vectors and shows some
vector properties. It presents the concept of colinear-
ity and dot product (inner product) of two vectors.
An application of these concepts is used to show how
to plot the altitude of a triangle with the dotprod

operator, as seen in fig. 1.

u:=1cm;

z1=origin; z2=(5u,-u); z3=(2u,2u);

draw z1--z2--z3--cycle;

dotlabel.llft(btex $A$ etex, z1);

dotlabel.lrt(btex $B$ etex, z2);

dotlabel.top(btex $C$ etex, z3);

(z2-z1) dotprod (z4-z3)=0;

z4=whatever[z1,z2];

draw z3--z4 dashed evenly;

dotlabel.llft(btex $H_C$ etex, z4);

Figure 1: Triangle ABC with relative altitude drawn
on side AB

The second chapter also shows the mathematics
behind the drawing of a regular polygon and the four
triangle centers: centroid, circumcenter, incenter and
orthocenter. In addition, it shows how to draw them
with MetaPost (figs. 2–6).

Chapter 3 discusses transformations in the plane,
as well as features that allow you to use them in Meta-
Post to draw various graphics. Some applications

Figure 2: Decagon of side 1 cm.

Figure 3: Medians and the centroid G of ABC.

Figure 4: Perpendicular bisectors and the
circumcenter O of ABC.

Figure 5: Altitudes and the orthocenter H of ABC.

Figure 6: Angle bisectors and the incenter I of ABC.

of these transformations are also presented for the
sketch of circles and ellipses.

A transformation in the plane R2 is a func-
tion T : R2 → R2, i.e, a correspondence that as-
sociates with each point P of the plane another
point P1 = T (P ) of the plane, called its image by
T . In MetaPost, transformations are performed us-
ing the transform type. With them, along with
the fullcircle command, you can draw circles and
ellipses, as shown in figures 7 and 8.

Figure 7: Circle C of center A and radius r

Figure 8: Ellipse E of center O, focuses F1 and F2 and
major axis 2a

Although it may seem obvious that using the
scaled, xscaled, yscaled, and shifted transforms
generates the circles and ellipses from the initial circle
fullcircle, the mathematics behind this concept
is accurate and, if I may say, particularly beautiful.
These results are exemplified with two interesting
problems of geometry: the nine-point circle and the
problem of tracing the ellipse tangent to the triangle
with one of its focuses an inner point of the triangle,
as shown in figures 9 and 10.

numeric u; u=1cm;

pair A,B,C,D,E,F,G,H,I,J,K,L,O,X;

A=(0,u);

B=(5u,u);

C=(u,10u);

D=.5[A,B];

E=.5[B,C];

F=.5[A,C];

G=whatever[C,B];

H=whatever[A,C];

I=whatever[B,A];

(A-G) dotprod (B-C) =

(B-H) dotprod (C-A) =

(C-I) dotprod (A-B) = 0;

X = whatever[A,G] = whatever[B,H];

J=.5[A,X]; K=.5[B,X]; L=.5[C,X];

(O-0.5(D+I)) dotprod (D-I) =

(O-0.5(G+K)) dotprod (G-K) = 0;

draw A--B--C--cycle;

draw fullcircle

scaled (2*abs(O-D)) shifted O;

Figure 9: Nine-point circle

Chapter 4 deals with the elaboration of graphs
of continuous functions defined in intervals of the line.
Initially a summary is made with some propositions
about continuous functions. Next, some MetaPost
techniques are outlined to sketch function graphs
through macros (fig. 11). The techniques in this
chapter discuss how to use a loop to calculate coor-
dinates of graph points, how to cut a piece of the
graph you want to see, how to use or create macros
for non-predefined math functions in MetaPost, and
how to plot graphs with asymptotes.

numeric u,a,b,c,theta; transform T;

pair A,B,C,F[],G[],O,P[]; path E;

u=1cm; A=(-u,-3u); B=(3u,0); C=(u,7u);

draw A--B--C--cycle; F1=(u,-u); F3=whatever[A,C];

F4=whatever[A,B]; F5=whatever[C,B];

(F3-F1) dotprod (A-C)= (F4-F1) dotprod (A-B)=

(F5-F1) dotprod (C-B)=0;

G1 = F1 rotatedaround (F3,180);

G2 = F1 rotatedaround (F4,180);

G3 = F1 rotatedaround (F5,180);

(F2-0.5(G2+G3)) dotprod (G2-G3) =

(F2-0.5(G1+G2)) dotprod (G1-G2) = 0;

P1=whatever[A,C]=whatever[F2,G1]; O=0.5[F1,F2];

2a=abs(F1-P1)+ abs(F2--P1); c=abs(F1--O); b=a+-+c;

theta=angle(F2-F1); T:= identity xscaled 2a

yscaled 2b rotated theta shifted O;

E:= fullcircle transformed T; draw E;

Figure 10: Ellipse inscribed in the triangle ABC with
one focus at F

Figure 11: Functions f(x) = 1 +
√

1− |x− 1|2 and
g(x) = 1 + arccos (1− |x|)− π (grayscaled for print)

For instance, the graph of figure 12 was sketched
using some of these techniques. First, some macros
are defined for the calculation of the cosine function.
Then the limits of the axes of the Cartesian plane
and the limits of the graph of the function are defined.
Since the function is defined in two disjoint intervals,
a loop is used to calculate 1000 points of each interval.
Then another loop is used to define the two smooth
curves through these points. The last step is to draw
these two curves respecting the preset limits.

def pi = arclength fullcircle enddef;

numeric radian; radian := 180/pi;

vardef cos primary x = (cosd(x*radian)) enddef;

numeric u, minx, maxx, miny, maxy; path q;

u:=0.8cm; minx:=-4; maxx:=4; miny:=-4; maxy:=4;

drawarrow ((minx,0)--(maxx,0)) scaled u;

drawarrow ((0,miny)--(0,maxy)) scaled u;

label.bot (btex $x$ etex, (maxx*u,0));

label.llft (btex $y$ etex, (0,maxy*u));

q:=((minx,miny)--(maxx,miny)--(maxx,maxy)--

(minx,maxy)--cycle) scaled 0.95;

numeric h,n; path p[];

n:=1000; a:=0.1; b:=maxx; h:=(b-a)/n;

for i=0 upto n: x[i]:=a+i*h;

y[i]:=(cos (10*x[i]))/x[i]; endfor;

p[1]=z[0] for j=1 upto n: ..z[j] endfor;

a:=minx; b:=-0.1; h:=(b-a)/n;

for i=0 upto n: x[i]:=a+i*h;

y[i]:=(cos (10*x[i]))/x[i]; endfor;

p[2]=z[0] for j=1 upto n: ..z[j] endfor;

pickup pencircle scaled 1bp;

draw (p1 cutafter q cutbefore q)

scaled u withcolor red;

draw (p2 cutafter q cutbefore q)

scaled u withcolor red;

Figure 12: Function f(x) =
cos 10x

x
(grayscaled for print)

Final considerations

MetaPost is a powerful tool for using mathematical
knowledge in the preparation of figures. The images
generated through this language have a satisfactory
professional level for what is expected of an academic
work. People who teach and research mathematics
can improve the presentation of their work and guide
their students in the application of acquired math-
ematical knowledge. Thus, I hope that this work
may, in a way, have contributed to the development
of mathematics teachers and, consequently, to the
development of their students.

I am part of a research group of the Institute
of Mathematics of the Federal University of Mato
Grosso do Sul, called “Fundamentals of Mathematics
for High School and the use of the language Meta-
Post”. This project proposes to provide material for
the high school curriculum using TEX and MetaPost.
We also intend to disseminate this language in the
academic environment of Brazil, since there is very
little work on MetaPost in Portuguese. If you would
like to collaborate, please contact us via my email
address.

